Dérivées partielles (Partie 1 de 2)

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2006. Le présent document est le premier d'une série de deux portant sur les sujets de la dérivée partielle et de la dérivée directionnelle. Chaque document transpose en Maple la présentation habituellement faite en classe de ces deux notions du calcul différentiel. Sur le plan mathématique, la compréhension des différentes étapes dans la définition de la dérivée partielle et directionnelle (orientée), exige, chez le lecteur, une maîtrise minimale de la géométrie vectorielle de la droite et du plan dans l'espace. Et à un niveau autre que purement mathématique, la transposition graphique en Maple de la notion de dérivée partielle est certainement l'occasion pour le lecteur d'enrichir ses compétences dans l'utilisation de Maple. Le lecteur tirera donc de ce document plusieurs éléments enrichissants en autant qu'il puisse en faire une lecture attentive et active.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> macro(grispâle = COLOR(RGB, .9523, .9523, .9523)):
  with(plots,contourplot,display,setoptions3d,spacecurve,textplot,
  textplot3d):
  with(plottools,disk,sphere,arrow):
  setoptions3d(axes=normal,style=patchnogrid, size=[300,300],
                lightmodel=none,
                axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8]);
```

Dérivées partielles... les définitions

Quelques traces à suivre ...

Soit le paraboloïde elliptique d'équation $z = 25 - \frac{x^2}{25} - \frac{y^2}{9}$.

```
> f:=(x,y)->25-1/25*x^2-1/9*y^2:
  'f'(x,y)=f(x,y);
```

$$f(x, y) = 25 - \frac{x^2}{25} - \frac{y^2}{9}$$

Traçons ce paraboloïde.

```
> Surface:=plot3d([x,y,f(x,y)],x=-30..30,y=-3*sqrt(25-1/25*x^2)..3*
  sqrt(25-1/25*x^2),\
                   style=patchcontour,color=turquoise):
  display(Surface,orientation=[55,65]);
```



Illustrons les différentes traces obtenues avec les intersections de cette surface avec des plans

– parallèles au plan *XOZ*

– parallèles au plan *YOZ*

– parallèles au plan *XOY*

Plans parallèles au plan *XOZ*. De tels plans ont des équations paramétriques de la forme

$x = r$
$y = k(constante)$
$z = s$; r, s ϵ R

```
> Surface:=plot3d([x,y,f(x,y)],x=-30..30,y=-3*sqrt(25-1/25*x^2)..3*
  sqrt(25-1/25*x^2),\
                   color=turquoise):
  Plan1:=plot3d([x,10,z],x=-30..30,z=0..25,color=plum):
  display([Surface,Plan1],orientation=[50,65]);
```



Illustrons ensuite quelques plans d'équations $y = k$ où $k \in [-13, -10, -5, 0, 5, 10, 13]$ sécants au paraboloïde elliptique.

```
> Plans_XOZ:=seq(display(Surface,plot3d([x,k,z],x=-30..30,z=0..30,
  color=plum),
                 spacecurve([t,k,f(t,k)],t=-25..25,color=
  "Niagara-8",thickness=4)),k=[-13,-10,-5,0,5,10,13]):
  display(Plans_XOZ,[Surface],insequence=true,orientation=[45,45],
  view=[-30..30,-15..15,0..30]);
```



Sélectionnez le graphique précédent et lancez l'animation avec le bouton "Play" de la barre de menu contextuel.

Faisons afficher seulement leur trace respective avec la surface du paraboloïde.

```
> Fig_1:=seq(spacecurve([t,k,f(t,k)],t=-25..25,color=navy,thickness=
  2),k=[-13,-10,-5,0,5,10,13]):
  display(Fig_1,view=[-30..30,-15..15,0..30]);
```



Plans parallèles au plan *YOZ*. Similairement, illustrons ensuite une série de tels plans d'équation $x = k$ où $k \in [20, 15, 10, 5, 0, -5, -10, -15, -20]$ avec le paraboloïde.

```
> Plans_YOZ:=seq(display(Surface,plot3d([k,y,z],y=-15..15,z=0..30,
  color=plum),
                 spacecurve([k,t,f(k,t)],t=-25..25,color=navy,
  thickness=2)),
                 k=[-20,-15,-10,-5,0,5,10,15,20]):
  display([Surface,Plans_YOZ],orientation=[45,45],insequence=true,
  view=[-30..30,-15..15,0..30]);
```



Lancez l'animation avec le bouton "Play".

Faisons afficher seulement leur trace respective avec la surface du paraboloïde.

```
> Fig_2:=seq(spacecurve([k,t,f(k,t)],t=-25..25,color=navy,thickness=
  2),k=[-20,-15,-10,-5,0,5,10,15,20]):
  display(Fig_2,view=[-30..30,-15..15,0..30]);
```



Plans parallèles au au plan *XOY*. Similairement, illustrons ensuite une série de tels plans d'équation $z = k$ où $k \in [23, 20, 15, 10, 5, 0]$ avec le paraboloïde.

Afin d'obtenir de beaux tracés des traces elliptiques, nous utiliserons des équations paramétiques plutôt que leur équation cartésienne.

$$x = 5\cos(t)\sqrt{25-k}$$
$$y = 3\sin(t)\sqrt{25-k}$$

```
> Plans_XOY:=seq(display(Surface,plot3d([x,y,k],x=-30..30,y=-15..15,
  color=plum),
            spacecurve([5*cos(t)*sqrt(25-k),3*sin(t)*sqrt(25-k),k],
  t=0..2*Pi,
            color=navy,thickness=2)),k=[23,20,15,10,5,0]):
  display([Surface,Plans_XOY],orientation=[45,45],insequence=true,
  view=[-30..30,-15..15,0..30]);
```



Lancez l'animation avec le bouton "Play".

Faisons afficher seulement leur trace respective avec la surface du paraboloïde.

```
> Fig_3:=seq(spacecurve([5*cos(t)*sqrt(25-k),3*sin(t)*sqrt(25-k),k],
  t=0..2*Pi,color=navy,thickness=2),k=[23,20,15,10,5,0]):
```

```
display(Fig_3);
```



Résumons. Superposons, dans un même graphique, toutes les différentes traces obtenues précedemment.

```
> display(Fig_||(1..3),view=[-30..30,-15..15,0..30]);
```



Dérivée partielle par rapport à la variable *x*.

Considérons de nouveau la surface paraboloïque elliptique d'équation $z = 25 - \frac{x^2}{25} - \frac{y^2}{9}$ mais en le limitant au premier octant.

```
> 'f'(x,y)=f(x,y);
```

$$f(x, y) = 25 - \frac{x^2}{25} - \frac{y^2}{9}$$

Illustrons alors cette surface dans le premier octant en accentuant le pourtour.

```
> L1:=spacecurve([t,sqrt((625*9-9*t^2)/25),0],t=0..25,linestyle=1,
  thickness=2,color=navy):
  L2:=spacecurve([t,0,f(t,0)],t=0..25,linestyle=1,thickness=2,color=
  navy):
  L3:=spacecurve([0,t,f(0,t)],t=0..15,linestyle=1,thickness=2,color=
  navy):
  Surface:=L||(1..3),plot3d([x,y,f(x,y)],x=0..30,y=0..3*sqrt(25
  -1/25*x^2),color=turquoise,transparency=0.2):
  display(Surface,orientation=[65,80],view=[-5..25,0..15,0..25]);
```



Superposons, à cette surface, le plan d'équation $y = 4$.

```
> PlanY:=plot3d([x,4,z],x=-5..25,z=0..25,color=plum):
  display([Surface,PlanY],orientation=[65,80],view=[-5..25,0..15,0.
  .25]);
```



Mettons en évidence la trace de ce plan sur la surface.

```
> Trace_PlanY:=spacecurve([t,4,f(t,4)],t=0..25,color=navy,thickness=
  3):
  display([Surface,Trace_PlanY],orientation=[65,80],view=[-5..25,0.
  .15,0..25]);
```



Considérons, sur cette trace, le point P1(5, 4, f(5, 4)).

```
> P1:=sphere([5,4,f(5,4)], 0.25,color=navy):
  Point1:=textplot3d([5,4+1,f(5,5)+2,"P1"],align={above,left},color=
  navy):
  PlanX1:=plot3d([x,4,z],x=-5..25,z=0..25,color=grispâle):
  Trace_PlanX:=spacecurve([t,4,f(t,4)],t=0..25,color=navy,thickness=
```

```
2):
Ligne1_XOY:=spacecurve([t,4,0],t=-5..25,color=grey,thickness=1):
Ligne1_YOZ:=spacecurve([-5,4,t],t=0..25,color=grey,thickness=1):
SegX_v1:=spacecurve([5,4,t*f(5,4)],t=0..1,linestyle=1,color=khaki)
:
Base:=P1,Point1,PlanX1,Trace_PlanX,Ligne1_XOY,Ligne1_YOZ:
display([Base,SegX_v1],orientation=[65,80],view=[-5..25,0..16,0.
.25]);
```



Nous allons maintenant illustrer le taux de variation partiel par rapport à x au point P1$(5, 4, f(5, 4))$.

Tout en maintenant y constant ($y = 4$), donnons un accroissement $h > 0$ à la variable x afin d'obtenir le point Q$(5 + h, 4, f(5 + h, 4))$.

Pour les besoins de l'illustration, prenons $h = 10$.

```
> h:=10:
  Q:=sphere([5+h,4,f(5+h,4)], 0.25,color=navy):
  SegX_v2:=spacecurve([5+h,4,t*f(5+h,4)],t=0..1,linestyle=1,color=
  khaki):
  PointQ:=textplot3d([5+h,4.5,f(5+h,4)+1.5,"Q"],align={above,left},
  color=navy):
  display([Base,Q,SegX_v1,SegX_v2,PointQ],orientation=[65,80],view=
  [-5..25,0..16,0..25]);
```



Traçons la sécante passant par les points P1$(5, 4, f(5, 4))$ et Q$(5 + h, 4, f(5 + h, 4))$.

Soit P1P2= (5+h-5,4-4, f(5+h,4)- f(5,4)) un vecteur directeur de cette sécante. Ainsi, des équations paramétriques de cette sécante sont données par

$x = 5 + t\,h$
$y = 4$
$z = \mathrm{f}(5,4) + t\,(\mathrm{f}(5+h,4) - \mathrm{f}(5,4))$; $t \in$.

```
> SécanteX1:=spacecurve([5+t*h,4,f(5,4)+t*(f(5+h,4)-f(5,4))],t=-3.
  .3,linestyle=1,color=magenta):
  display([Base,Q,SegX_v1,SegX_v2,PointQ,SécanteX1],orientation=[60,
  80],view=[-5..25,0..16,0..25]);
```



Peaufinons cette illustration en complétant les éléments graphiques contribuant à mette en évidence la sécante passant par les points P1 et Q.

```
> SegX_v1_bis:=spacecurve([5,4,t*f(5,4)],t=0..f(15,4)/f(5,4),
  linestyle=1,color=khaki):
  SegX_y2:=spacecurve([t,4,0],t=0..15,linestyle=1,color=grey):
  LigneX_1:=spacecurve([15-t,4,f(15,4)],t=0..10,linestyle=1,color=
  magenta):
  LigneX_2:=spacecurve([5,4,f(15,4)+t],t=0..f(5,4)-f(15,4),
  linestyle=1,color=magenta):
  display([Base,Q,SegX_v1_bis,SegX_v2,PointQ,SécanteX1],orientation=
  [65,60],LigneX_1,LigneX_2,view=[-5..25,0..16,0..25]);
```



Dans le plan $y = 4$, nous ferons tendre le point Q vers le point P1 en faisant tendre h vers 0. Ainsi, cela nous donnera le taux de variation partiel par rapport à x au point P1$(5, 4, f(5, 4))$.

```
> h:='h':
  Limit(('f'(5+h,4)-'f'(5,4))/h,h=0,right)=Limit(normal((f(5+h,4)-f
  (5,4))/h),h=0,right);
  ``=limit((f(5+h,4)-f(5,4))/h,h=0,right);
```

$$\lim_{h \to 0+} \frac{f(5+h,4) - f(5,4)}{h} = \lim_{h \to 0+} \left(-\frac{h}{25} - \frac{2}{5} \right)$$

$$= -\frac{2}{5}$$

L'interprétation graphique de la valeur $-\frac{2}{5}$ est la pente de la tangente **à la trace** de la surface f et du plan $y = 4$ au point P1.

Traçons alors cette tangente. Pour obtenir un vecteur directeur de cette tangente, il suffit de prendre le vecteur $(1, 0, -\frac{2}{5})$. Ainsi, des équations paramétriques de cette tangente sont données par:

$$x = 5 + t$$
$$y = 4$$
$$z = f(5,4) - t\,\frac{2}{5};\ t \in \ .$$

De plus, donnons au graphique une orientation montrant clairement que toute cette approche est réalisée, somme toute, dans un plan tout comme pour la dérivée d'une fonction d'une seule variable.

```
> TangenteX1:=spacecurve([5+t,4,f(5,4)+t*(-2/5)],t=-10..10,\
            linestyle=1,thickness=2,color=orange):
  display([Base,Q,SegX_v1_bis,SegX_v2,PointQ,SécanteX1,LigneX_1,
  LigneX_2,TangenteX1],\
         orientation=[65,60],view=[-5..25,0..16,0..25]);
```



Montrons, dans un même graphique, le tracé de cette première tangente au point P1 de la surface selon la direction du plan parallèle à l'axe des x, soit le plan $y = 4$. La pente de cette tangente est la ***valeur de la dérivée partielle par rapport à x évaluée avec l'abscisse et l'ordonnée*** du point P1.

```
> Graphique_1:=Base,SegX_v1,TangenteX1:
  display([Surface,Graphique_1],orientation=[65,60],view=[-5..25,0.
  .16,0..25]);
```



```
> display([Surface,P1,Point1,Trace_PlanX,TangenteX1,SegX_v1],\
          orientation=[65,80],view=[-5..25,0..16,0..25]);
```



Comme deuxième exemple, illustrons la valeur du taux de variation partiel par rapport à x au point P2(10, 12, f(10,12)).

Obtenons ce taux en évaluant directement la dérivée partielle par rapport à x avec $x = 10$ et $y = 12$. Utilisons l'opérateur de dérivation D, en lui spécifiant le nombre 1 en indice, le résultat sera la dérivée partielle par rapport à la première variable utilisée lors de la définition de la fonction f. Notons que le résultat est une fonction.

```
> fx:=D[1](f);
```

$$fx := (x, y) \mapsto -\frac{2 \cdot x}{25}$$

Ainsi, le taux de variation partiel m par rapport à x au point P2(10,12,f(10,12)) est:

```
> m=fx(10,12);
```

$$m = -\frac{4}{5}$$

La pente de la tangente au point P2 de la trace de la surface f et du plan d'équation $y = 12$ est égale à $-\frac{4}{5}$.

Illustrons ce taux en traçant la tangente au point P2 de cette trace.

Pour obtenir un vecteur directeur de cette tangente, il suffit de prendre le vecteur $\left(0, 1, -\frac{4}{5}\right)$. Ainsi, des équations paramétriques de cette tangente sont données par:

$x = 10 + t$

$y = 12$

$z = \mathrm{f}(10, 12) - t\,\frac{4}{5}$; $t \in$.

```
> P2:=sphere([10,12,f(10,12)], 0.25,color=navy):
  Point2:=textplot3d([10,12+1,f(10,12)+2,"P2"],align={above,left},
  color=navy):
  PlanP2:=plot3d([x,12,z],x=-5..25,z=0..15,color=grispâle):
  Trace_PlanP2:=spacecurve([t,12,f(t,12)],t=0..25,color=navy,
  thickness=2):
  Ligne2_XOY:=spacecurve([t,12,0],t=-5..25,color=grey,thickness=1):
  Ligne2_YOZ:=spacecurve([-5,12,t],t=0..15,color=grey,thickness=1):
  SegX_xx1:=spacecurve([10,t,0],t=0..12,linestyle=1,color=grey):
  SegX_yy1:=spacecurve([t,12,0],t=0..10,linestyle=1,color=grey):
```

```
SegX_vv1:=spacecurve([10,12,t*f(10,12)],t=0..1,linestyle=1,color=
khaki):
TangenteX2:=spacecurve([10+t,12,f(10,12)+t*(-4/5)],t=-5..10,\
             linestyle=1,thickness=2,color=orange):
Graphique_2:=P2,Point2,SegX_vv1,PlanP2,Trace_PlanP2,\
               Ligne2_XOY,Ligne2_YOZ,TangenteX2:
display([Graphique_1,Graphique_2],orientation=[50,65],view=[-5.
.25,0..16,0..25]);
```




```
> display([Surface,P1,Point1,Trace_PlanX,TangenteX1,\
                  P2,Point2,Trace_PlanP2,TangenteX2,SegX_vv1,
SegX_v1],\
                  orientation=[65,60],view=[-5..25,0..16,0..25]);
```



Dérivée partielle par rapport à la variable *y*.

Dans le cas où le lecteur commencerait par cette section-ci, recréons la surface paraboloïque elliptique d'équation $z = 25 - \frac{x^2}{25} - \frac{y^2}{9}$ limitée au premier octant. Ne pas oublier, dans ce cas, d'exécuter les requêtes de la section Initialisation.

```
> f:=(x,y)->25-1/25*x^2-1/9*y^2:
'f'(x,y)=f(x,y);
L1:=spacecurve([t,sqrt((625*9-9*t^2)/25),0],t=0..25,linestyle=1,
color=navy):
L2:=spacecurve([t,0,f(t,0)],t=0..25,linestyle=1,color=navy):
L3:=spacecurve([0,t,f(0,t)],t=0..15,linestyle=1,color=navy):
```

```
Surface:=L||(1..3),plot3d([x,y,f(x,y)],x=0..30,y=0..3*sqrt(25
-1/25*x^2),color=turquoise,transparency=0.2,view=[-5..25,-5..20,0.
.25]):
display(Surface,orientation=[65,60]);
```

$$f(x,y) = 25 - \frac{x^2}{25} - \frac{y^2}{9}$$



Superposons, à cette surface, le plan d'équation $x = 5$.

```
> PlanY:=plot3d([5,y,z],y=-5..205,z=0..25,color=plum):
  display([Surface,PlanY],orientation=[50,65],view=[-5..25,-5..20,0.
  .25]);
```



Mettons en évidence la trace de ce plan sur la surface.

```
> Trace_PlanY:=spacecurve([5,t,f(5,t)],t=0..15,color=navy,thickness=
  3):
  display([Surface,Trace_PlanY],orientation=[50,65],view=[-5..25,0.
  .15,0..25]);
```



Considérons, sur cette trace, le point P1(5, 4, f(5, 4)).

```
> P1 := sphere([5,4,f(5,4)], 0.25,color=navy):
  PointP1:=textplot3d([6,4,f(5,5)+2,"P1"],align={above,left},color=
  navy):
  SegY_v1:=spacecurve([5,4,t*f(5,4)],t=0..1,linestyle=1,color=khaki)
  :
  PlanY1:=plot3d([5,y,z],y=-5..25,z=0..25,color=grispâle):
  Trace_PlanY1:=spacecurve([5,t,f(5,t)],t=0..15,color=navy,
  thickness=1):
  Ligne2_XOY:=spacecurve([5,t,0],t=-5..15,color=grey,thickness=1):
  Ligne2_XOZ:=spacecurve([5,-5,t],t=0..25,color=grey,thickness=1):
  Base:=P1,PointP1,PlanY1,Trace_PlanY1,Ligne2_XOY,Ligne2_XOZ:
  display([Base,SegY_v1],orientation=[20,60],view=[0..25,-5..15,0.
  .25]);
```



Nous allons maintenant illustrer le taux de variation partiel par rapport à y au point P1(5, 4, $f(5, 4)$).

Tout en maintenant x constant ($x = 5$), donnons un accroissment $h > 0$ à la variable y afin d'obtenir le point Q(5, 4 + h, f(5, 4 + h)).

Pour les besoins de l'illustration, prenons $h = 6$.

```
> h:=6:
  Q:=sphere([5,4+h,f(5,4+h)], 0.25,color=navy):
  PointQ:=textplot3d([6,11.5,f(5,10)+1,"Q"],align={above,left},
  color=navy):
  SegY_v2:=spacecurve([5,4+h,t*f(5,4+h)],t=0..1,linestyle=1,color=
  khaki):
  display([Base,PointQ,SegY_v1,SegY_v2],orientation=[20,60],view=[0.
  .25,-5..15,0..25]);
```



Traçons la sécante passant par les points P1(5, 4, $\mathrm{f}(5,4)$) et Q(5, $4+h$, $\mathrm{f}(5,4+h)$).
Soit P1Q= (5-5,(4+h)-4, f(5,4+h)- f(5,4)) un vecteur directeur de cette sécante. Ainsi, des équations paramétriques de cette sécante sont données par

$x=5$

$y=4+t\,h$

$z=\mathrm{f}(5,4)+t\,(\mathrm{f}(5,4+k)-\mathrm{f}(5,4))$; $t\in$.

```
> LigneY_1:=spacecurve([5,t,f(5,10)],t=4..10,linestyle=1,color=
  magenta):
  LigneY_2:=spacecurve([5,4,t],t=f(5,10)..f(5,4),linestyle=1,color=
  magenta):
  SegY_v1_bis:=spacecurve([5,4,t],t=0..f(5,10),linestyle=1,color=
  khaki):
  SécanteY1:=spacecurve([5,4+h*t,f(5,4)+t*(f(5,4+h)-f(5,4))],t=-0.5.
  .1.5,\
             linestyle=1,color=magenta):
  display([Base,SegY_v1_bis,SegY_v2,PointQ,LigneY_1,LigneY_2,
  SécanteY1],\
          orientation=[20,60],view=[0..25,-5..15,0..25]);
```



Dans le plan $x=5$, nous ferons tendre le point Q vers le point P1 en faisant tendre h vers 0. Ainsi, cela nous donnera le taux de variation partiel par rapport à y au point P1(5, 4, $\mathrm{f}(5,4)$).

```
> h:='h':
  Limit(('f'(5,4+h)-'f'(5,4))/h,h=0,right)=Limit(normal((f(5,4+h)-f
  (5,4))/h),h=0,right);
  ``=limit((f(5,4+h)-f(5,4))/h,h=0,right);
```

$$\lim_{h\to 0+}\frac{f(5,4+h)-f(5,4)}{h}=\lim_{h\to 0+}\left(-\frac{h}{9}-\frac{8}{9}\right)$$

$$= -\frac{8}{9}$$

L'interprétation graphique de la valeur $-\frac{8}{9}$ est donc la pente de la tangente à la trace de la fonction f et du plan $x = 5$ au point P1.
Traçons alors cette tangente. Pour obtenir un vecteur directeur de cette tangente, il suffit de prendre le vecteur $\left(0, 1, \frac{8}{9}\right)$. Ainsi, des équations paramétriques de cette tangente sont données par:

$$x = 5$$
$$y = 4 + t$$
$$z = \mathrm{f}(5, 4) - t\,\frac{8}{9}\,;\ t \in \ .$$

De plus, donnons au graphique une orientation montrant claitement que toute cette approche est réalisée dans un plan tout comme pour la dérivée d'une fonction d'une seule variable.

```
> TangenteY1:=spacecurve([5,4+t,f(5,4)+t*(-8/9)],t=-5..5,\
              linestyle=1,thickness=2,color=orange):
  display([Base,SegY_v1_bis,SegY_v2,PointQ,LigneY_1,LigneY_2,
  SécanteY1,TangenteY1],\
           orientation=[0,90],view=[0..25,-5..15,0..25]);
```



Montrons, dans un seul graphique, le tracé de cette première tangente au point P1 de la surface selon la direction du plan parallèle à l'axe des y, soit le plan $x = 5$. La pente de cette tangente est la ***valeur de la dérivée partielle par rapport à*** y ***évaluée avec l'abscisse et l'ordonnée*** du point P1.

```
> Graphique_1:=Base,SegY_v1,TangenteY1:
  display([Surface,Graphique_1],orientation=[20,60],view=[-5..25,-5.
  .15,0..25]);
```



```
> Fig_1:=Surface,P1,Point1,SegY_v1,Trace_PlanY1,TangenteY1:
  display(Fig_1,orientation=[20,60],view=[-5..25,0..15,0..25]);
```



Comme deuxième exemple, illustrons la valeur du taux de variation partiel par rapport à y au point P3(15, 8, f(15,8)).

Obtenons ce taux en évaluant directement la dérivée partielle par rapport à y avec $x = 15$ et $y = 8$. Utilisons l'opérateur de dérivation D. En lui spécifiant le nombre 2 en indice, le résultat sera la dérivée partielle par rapport à la seconde variable utilisée lors de la définition de la fonction f. Notons encore que le résultat est une fonction.

```
> fy:=D[2](f);
```

$$fy := (x, y) \mapsto -\frac{2 \cdot y}{9}$$

Ainsi, le taux de variation partiel m par rapport à y au point P3(15,8,f(15,8)) est:

```
> m=fy(15,8);
```

$$m = -\frac{16}{9}$$

La pente de la tangente au point P3 de la trace de la surface f et du plan d'équation $x = 15$ est égale à $-\frac{16}{9}$. Illustrons ce taux en traçant la tangente au point P3 de cette trace.

Pour obtenir un vecteur directeur de cette tangente, il suffit de prendre le vecteur $\left(0, 1, -\frac{16}{9}\right)$. Ainsi, des équations paramétriques de cette tangente sont données par:

$$x = 15$$
$$y = 8 + t$$
$$z = \mathrm{f}(15, 8) - t\,\frac{16}{9}\ ;\ t \in\ .$$

```
> Trace_PlanY2:=spacecurve([15,t,f(15,t)],t=0..12,color=navy,
  thickness=1):
  P3:=sphere([15,8,f(15,8)],0.25,color=navy):
  SegY_vv1:=spacecurve([15,8,t*f(15,8)],t=0..1,linestyle=1,color=
  khaki):
  PointP3:=textplot3d([15,9.5,f(15,8)+1,"P3"],align={above,left},
  color=navy):
  TangenteY2:=spacecurve([15,8+t,f(15,8)+t*(-16/9)],t=-3..3,\
               linestyle=1,thickness=2,color=orange):
  Fig_2:=Surface,P3,PointP3,Trace_PlanY2,TangenteY2,SegY_vv1:
  display([Fig_1,Fig_2],orientation=[25,60],view=[-5..25,0..15,0.
```

```
.25]);
```



Plan tangent

Visualisons les deux tangentes simultanément au point P1.

```
> display([Surface,P1,PointP1,Trace_PlanX,Trace_PlanY1,TangenteX1,
  TangenteY1],orientation=[20,60],view=[0..25,0..15,0..28]);
```



Grâce à ces deux tangentes, il nous est facile de voir qu'il est possible de tracer un plan tangent à cette surface au point P1. Il suffit de considérer deux vecteurs directeurs respectifs des deux tangentes pour bâtir des équations paramétriques de ce plan tangent.

Soit les vecteurs directeurs suivants: $v_1 = \left(1, 0, \frac{4}{5}\right)$ et $v_2 = \left(0, 1, -\frac{8}{9}\right)$. En considérant le point P1(5,4,f(5,4)), nous avons des équations paramétriques suivantes:

$x = 5 + r$

$y = 4 + s$

$z = f(5,4) - r\,\frac{2}{5} - s\,\frac{8}{9}$; $r, s \in \mathbb{R}$.

```
> Plan_Tangent:=plot3d([5+r,4+s,f(5,4)-2*r/5-8*s/9],r=-6..10,s=-6.
  .5,color=pink):
  display(Surface,P1,PointP1,Plan_Tangent,TangenteX1,TangenteY1,
  orientation=[-30,75],\
          view=[0..25,0..16,0..28],orientation=[-24,81,-3],scaling=
  unconstrained);
```